

# NAG Library Routine Document

## F07VGF (DTBCON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07VGF (DTBCON) estimates the condition number of a real triangular band matrix.

### 2 Specification

```
SUBROUTINE F07VGF (NORM, UPLO, DIAG, N, KD, AB, LDAB, RCOND, WORK, IWORK, INFO)
INTEGER N, KD, LDAB, IWORK(N), INFO
REAL (KIND=nag_wp) AB(LDAB,*), RCOND, WORK(3*N)
CHARACTER(1) NORM, UPLO, DIAG
```

The routine may be called by its LAPACK name *dtbcon*.

### 3 Description

F07VGF (DTBCON) estimates the condition number of a real triangular band matrix  $A$ , in either the 1-norm or the  $\infty$ -norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that  $\kappa_\infty(A) = \kappa_1(A^T)$ .

Because the condition number is infinite if  $A$  is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine computes  $\|A\|_1$  or  $\|A\|_\infty$  exactly, and uses Higham's implementation of Hager's method (see Higham (1988)) to estimate  $\|A^{-1}\|_1$  or  $\|A^{-1}\|_\infty$ .

### 4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

### 5 Parameters

- 1: NORM – CHARACTER(1) *Input*  
*On entry:* indicates whether  $\kappa_1(A)$  or  $\kappa_\infty(A)$  is estimated.  
 NORM = '1' or 'O'  
 $\kappa_1(A)$  is estimated.  
 NORM = 'I'  
 $\kappa_\infty(A)$  is estimated.  
*Constraint:* NORM = '1', 'O' or 'I'.
- 2: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies whether  $A$  is upper or lower triangular.  
 UPLO = 'U'  
 $A$  is upper triangular.

- UPLO = 'L'  
*A* is lower triangular.  
 Constraint: UPLO = 'U' or 'L'.
- 3: DIAG – CHARACTER(1) *Input*  
*On entry:* indicates whether *A* is a nonunit or unit triangular matrix.  
 DIAG = 'N'  
*A* is a nonunit triangular matrix.  
 DIAG = 'U'  
*A* is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.  
 Constraint: DIAG = 'N' or 'U'.
- 4: N – INTEGER *Input*  
*On entry:* *n*, the order of the matrix *A*.  
 Constraint:  $N \geq 0$ .
- 5: KD – INTEGER *Input*  
*On entry:*  $k_d$ , the number of superdiagonals of the matrix *A* if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.  
 Constraint:  $KD \geq 0$ .
- 6: AB(LDAB,\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array AB must be at least  $\max(1, N)$ .  
*On entry:* the *n* by *n* triangular band matrix *A*.  
 The matrix is stored in rows 1 to  $k_d + 1$ , more precisely,  
   if UPLO = 'U', the elements of the upper triangle of *A* within the band must be stored with element  $A_{ij}$  in  $AB(k_d + 1 + i - j, j)$  for  $\max(1, j - k_d) \leq i \leq j$ ;  
   if UPLO = 'L', the elements of the lower triangle of *A* within the band must be stored with element  $A_{ij}$  in  $AB(1 + i - j, j)$  for  $j \leq i \leq \min(n, j + k_d)$ .  
 If DIAG = 'U', the diagonal elements of *A* are assumed to be 1, and are not referenced.
- 7: LDAB – INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F07VGF (DTBCON) is called.  
 Constraint:  $LDAB \geq KD + 1$ .
- 8: RCOND – REAL (KIND=nag\_wp) *Output*  
*On exit:* an estimate of the reciprocal of the condition number of *A*. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**, *A* is singular to working precision.
- 9: WORK(3 × N) – REAL (KIND=nag\_wp) array *Workspace*
- 10: IWORK(N) – INTEGER array *Workspace*
- 11: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value  $\rho$ , and in practice is nearly always less than  $10\rho$ , although examples can be constructed where RCOND is much larger.

## 8 Parallelism and Performance

F07VGF (DTBCON) is not threaded by NAG in any implementation.

F07VGF (DTBCON) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

A call to F07VGF (DTBCON) involves solving a number of systems of linear equations of the form  $Ax = b$  or  $A^T x = b$ ; the number is usually 4 or 5 and never more than 11. Each solution involves approximately  $2nk$  floating-point operations (assuming  $n \gg k$ ) but takes considerably longer than a call to F07VEF (DTBTRS) with one right-hand side, because extra care is taken to avoid overflow when  $A$  is approximately singular.

The complex analogue of this routine is F07VUF (ZTBCON).

## 10 Example

This example estimates the condition number in the 1-norm of the matrix  $A$ , where

$$A = \begin{pmatrix} -4.16 & 0.00 & 0.00 & 0.00 \\ -2.25 & 4.78 & 0.00 & 0.00 \\ 0.00 & 5.86 & 6.32 & 0.00 \\ 0.00 & 0.00 & -4.82 & 0.16 \end{pmatrix}.$$

Here  $A$  is treated as a lower triangular band matrix with one subdiagonal. The true condition number in the 1-norm is 69.62.

### 10.1 Program Text

```

Program f07vgfe
!      F07VGF Example Program Text
!
!      Mark 25 Release. NAG Copyright 2014.
!
!      .. Use Statements ..
!      Use nag_library, Only: dtbcon, nag_wp, x02ajf
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Integer, Parameter          :: nin = 5, nout = 6
!      Character (1), Parameter    :: diag = 'N', norm = '1'
!      .. Local Scalars ..
!      Real (Kind=nag_wp)          :: rcond

```

```

Integer                                :: i, info, j, kd, ldab, n
Character (1)                          :: uplo
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable        :: ab(:,,:), work(:)
Integer, Allocatable                    :: iwork(:)
! .. Intrinsic Procedures ..
Intrinsic                               :: max, min
! .. Executable Statements ..
Write (nout,*) 'F07VGF Example Program Results'
! Skip heading in data file
Read (nin,*)
Read (nin,*) n, kd
ldab = kd + 1
Allocate (ab(ldab,n),work(3*n),iwork(n))

! Read A from data file

Read (nin,*) uplo
If (uplo=='U') Then
  Do i = 1, n
    Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
  End Do
Else If (uplo=='L') Then
  Do i = 1, n
    Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
  End Do
End If

! Estimate condition number
! The NAG name equivalent of dtbcon is f07vgf
Call dtbcon(norm,uplo,diag,n,kd,ab,ldab,rcond,work,iwork,info)

Write (nout,*)
If (rcond>=x02ajf()) Then
  Write (nout,99999) 'Estimate of condition number =', &
    1.0E0_nag_wp/rcond
Else
  Write (nout,*) 'A is singular to working precision'
End If

99999 Format (1X,A,1P,E10.2)
End Program f07vgfe

```

## 10.2 Program Data

```

F07VGF Example Program Data
  4  1                                :Values of N and KD
  'L'                                :Value of UPLO
-4.16
-2.25    4.78
          5.86    6.32
          -4.82    0.16    :End of matrix A

```

## 10.3 Program Results

F07VGF Example Program Results

Estimate of condition number = 6.96E+01

---